# $Id$
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=libmng
pkgver=2.0.3
pkgrel=1
pkgdesc="A collection of routines used to create and manipulate MNG format graphics files"
arch=('i686' 'x86_64')
url="http://www.libmng.com/"
license=('custom')
depends=('lcms2')
source=(http://downloads.sourceforge.net/sourceforge/${pkgname}/${pkgname}-${pkgver}.tar.xz{,.asc})
sha1sums=('0f141482ffcef6f8cd4413f945a59310ac2e49af'
          'SKIP')
validpgpkeys=('8048643BA2C840F4F92A195FF54984BFA16C640F')

build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -D -m644 LICENSE "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}
